
News 
The latest version of Spike2 is 5.09, which is available to download from
registered v5 users. This revision includes new script commands for IIR
additions to the clustering functions that now allow you to view the K Me
clustering processes, choose which principal components to display and
visualisation of clusters. 

Version 5.10 is in the works, and will include a multi-window view for on
channels in a data file. You will be able to set the number of spikes to o
to display.  

A full list of new features is available from the updates section of the we

A demonstration version of the current software is also available here. 

Version 4.21 of Spike2 is downloadable for registered v4 users. 

Version 3 of Signal is now available. This version implements many new
patch clamp analysis and expands on existing features with new improv
auto-average measurements. 

Version 3.01 of Signal is nearing release. This will include functions allo
be overdrawn.  

A demonstration version of Signal version 3 is also available here. 

Version 2.15 of Signal is downloadable for registered v2 users.  

Spike2  
Q. I would like to be able to choose which three principal compone

clustering window. 

A. The latest Spike2 version 5 includes a number of additional fun
window to enable better visualisation of the data. You can set th
clustering dialog from the Change colours menu, as well as ma
functions: 

 Select Principal components 

 In response to the above user request, you can now choose wh
display in the clustering window using the Select Principal Com

menu. You can force the components in order to ensure that the X component is the lowest and Z 
the highest, and cause the cluster window to update on every change made. The Reset button 
always selects the first three principal components.  

  
 Display Jiggle 

 There is a new setting in the cluster window View menu to “jiggle” the display around the current 
view angle. This means that, after rotating the display to get the best view of the clusters, you can 
apply a small jitter to help visualize the clusters in three dimensions.  

 View along axis 

 There is another new command in the View menu that enables you to directly view the cluster 
data along the X, Y or Z axis.  

  
  



 Display clustering process 

 When using the K Means and Normal Mixture clustering functions, you can follow the clustering 
process in detail. You can also break out of a long clustering operation. 

   

Signal 

Q. I would like to overdraw selected frames during sampling with Signal. To start with I would just like 
to be able to specify a single frame for overdraw, and be able to update this as required. 

A. Signal version 3 allows you to overdraw nominated frames during sampling. The buffer frame has 
now been added to the frame display list, meaning that you can copy a selected frame to the 
buffer during sampling. Simply select Buffer from the frame display list and, once you have copied 
your nominated frame to the buffer, this can be overdrawn on every new sweep. 

 
     

 Signal version 3.01 will expand on this by adding a last n frames filed option to the frame display 
list. This will enable you to overdraw a selected number of the last frames sampled, and will 
update appropriately as new data is added.  

 

Scripts: Spike2  

Q. Do you have any example scripts that show how to use the interactive dialog commands in Spike2 
version 5? 

A. The attached IIRMake.s2s script shows an example of using a dialog that responds to user 
interaction by enabling and disabling various dialog fields in response to any changes made. This 
script also shows how to use tooltips for dialog items. Moving the cursor over the dialog text will 
show a tooltip describing the field setting. 

 The script can be used to write example IIR filter commands. The user specifies filter settings in a 
generated dialog, and the script then writes the code for the correct script functions. These can 
then be used in your own scripts. The IIR commands are new in the latest Spike2 version 5, and 
you can use them to generate and apply Infinite Impulse Response filters to data that is held in 
arrays of real numbers. You can create digital filters that are modelled on Butterworth, Bessel, 
Chebyshev type 1 and type 2 highpass, lowpass, bandpass and bandstop filters. There are also 
commands to create digital resonators and notch filters. 

 When compared to FIR filters, IIR filters have the advantage of being able to generate much 
steeper edges and narrower notches, and they are causal which means that the filter output is 
only affected by current and previous data. However, they also have the disadvantage that they 
impose a phase shift on the data that varies with frequency. This means that they do not preserve 
the shape of a waveform, in particular, the position of peaks and troughs will change by a small 
amount.  

 The IIR commands in Spike2 also allow you to check the impulse, step, frequency and phase 
response of any generated filters, and we strongly recommend that you do so before using any of 


'$IIRMake|Example script that writes code for IIR commands based on dialog settings

const lo:= 0.0002;		'The minimum low pass setting for filters
const hi:= 0.4998;		'The minimum high pass setting for filters

var filtName%,filtType%;		'Variables returned from dialog holding filter name to use and type
var order%,notch%,reson%;		'Variables holding filter settings
var rate,cFreq1,cFreq2,ripple,nFreq,qFact;	'
var ok%;								'Returns 1 if OK pressed in dialog
var gFloat%[20];              'global for floating window states
var filtList$[4];					'Array to hold filter names
var v%,d%,f%,c%;		'variables to hold dialog field return values

filtList$[0]:="Butterworth";
filtList$[1]:="Bessel";
filtList$[2]:="Chebyshev type 1";
filtList$[3]:="Chebyshev type 2";

if App(-1)<509 then Message("This script needs at least Spike2 version 5.09"); halt endif;	'Checks for the latest version of Spike2

HideAll();			'Hide toolbar and script window

repeat		'If settings are not suitable, the dialog repeats
	DlgCreate("IIR Filter settings");
	DlgGroup("Select Filter Type",1,1,47,4);
	DlgList(1,"",filtList$[],4,4,2);
	DlgList(2,"","Lowpass|Highpass|Bandpass|Bandstop",4,28,2);
	DlgGroup("Data to be filtered",1,5.5,47,3);
	DlgReal(6,"Sample Rate (Hz)|The sample rate of the data you want to filter",100,50000,20,6.5,1);
	DlgGroup("Filter Characteristics",1,9,47,6);
	DlgCheck(3,"Notch",3,3);
	DlgCheck(4,"Resonator",3,4);
	DlgInteger(5,"Set Filter Order|For Bandpass and Bandstop filters, the Order used is twice this setting",1,10,22,10,1);

	DlgReal(7,"Edge Frequency 1 (Hz)",1,25000,22,11.2,1);
	DlgReal(8,"Edge Frequency 2 (Hz)",1,25000,22,12.4,1);
	DlgReal(9,"Ripple (dB)|This is the desired pass band ripple for type 1, or the cut in the stop band for type 2",1,100,22,13.6,1);
	DlgGroup("Notch and Resonator settings",1,15.5,47,3.5);
	DlgReal(10,"Notch Freq. (Hz)",1,25000,22,16.5,1);
	DlgReal(11,"q-Factor|The higher the q value, the narrower the notch. If unsure, try 100 as a starting value",10,1000,22,17.7,1);
	DlgAllow(1023,0,DlgSettings%);
	ok%:=DlgShow(filtName%,filtType%,notch%,reson%,order%,rate,cFreq1,cFreq2,ripple,nFreq,qFact);

	if ok%=0 then RestoreAll();halt;endif;		'If user presses cancel, quit script

	if ok%>0 then 				'If user presses OK
		if d%=1 or f%=1 then
			nFreq:= nFreq/rate;	 
			cFreq1:=lo;
			cFreq2:=lo;
		else;
		
		if d%=0 and f%=0 then
		cFreq1:= cFreq1/rate;	'Calculate used frequencies as a fraction of sample rate
			if v%=2 or v%=3 then
				cFreq2:= cFreq2/rate;	'These statements check for enabled fields, and only calculates frequencies for those
				else
				cFreq2:=lo;
			endif;
		nFreq:=lo;
		endif;
		endif;
	
		if cFreq1 > hi or cFreq1 < lo or cFreq2 > hi or cFreq2 < lo or nFreq > hi or nFreq < lo then		'If outside the acceptable range
			Message("The Edge/Notch frequencies must be less than the Nyquist frequency\n(half the sample rate), and greater than 0.0002 of the Sample Rate!");	'warn user
			ok%:=0;		'Reset ok% so that dialog repeats
		endif;
	endif;	
until ok%>0;		'until we have acceptable settings to proceed

WriteIIRCode%();	'Calls the function that writes the IIR code

'**************************************************************************************************************************
'									Script functions
'**************************************************************************************************************************

Func DlgSettings%(item%)	'This function monitors user changes to the dialog so that only relevant fields can be selected

v%:= DlgValue(2);		'Get type of filter
d%:= DlgValue(3);		'Check if Notch selected
f%:= DlgValue(4);		'Or Resonator...
c%:= DlgValue(1);		'Get filter name

DlgEnable(0,8,9);

docase
	case d%=1 then			'If Notch filter selected
		DlgValue(4,0);		'Clear resonator
		DlgValue(3,1);		'Set Notch
		DlgEnable(0,1,2,5,7,8);		'Set dialog to initial state
		DlgEnable(1,10,11);

	case f%=1 then			'If Resonator selected
		DlgValue(3,0);		'Clear Notch
		DlgValue(4,1);		'Set Resonator
		DlgEnable(0,1,2,5,7,8);		'Set dialog to initial state
		DlgEnable(1,10,11);

	case d%=0 and f%=0 then		'If neither selected, enable all standard fields
		DlgEnable(1,1,2,5,7);			'Enable Filter name field
		DlgEnable(0,10,11);			'Disable Notch.. & Resonator settings
		
		if v%=2 or v%=3  then	'If Bandpass or Bandstop filter selected..
			DlgEnable(1,8);		'..Enable Edge frequency 2 field
		endif;
		if c%=2 or c%=3 then		'If either Chebyshev filter selected..
			DlgEnable(1,9);		'..Enable ripple field
	 	endif;
endcase;
return 1;
end;

Func WriteIIRCode%()
var script%;		'Variable to hold view handle of new script file
script%:=FileNew(3);		'Open new script file

Print("var filtName%:= %d;\t",filtName%);				'Print standard variables using settings from dialog
Print("'Filter type = %s\n",filtList$[filtName%]);	 
Print("var order% := %d;\t'Filter order\n", order%);

if filtName%=2 then		'If Chebyshev type 1..
	Print("var ripple := %f;\t'Pass band ripple\n",ripple);
endif;
if filtName%=3 then		'If Chebyshev type 2..
	Print("var ripple := %f;\t'Stop band cut\n",ripple);
endif;

docase
	case notch%=1 or reson%=1 then		'If filter selected is Notch or Resonator..
	 	Print("var nFreq := %f;\t'Notch Frequency\n",nFreq);		
		Print("var qFact := %f;\t'q-factor for filter\n",qFact);		'Print special variables
		if notch%=1 then
			Print("\nIIRNotch(0,nFreq,qFact);\t'Generated Notch filter");	'Specify notch or Resonator
		else
			Print("\nIIRReson(0,nFreq,qFact);\t'Generated Resonator filter");
		endif;

	case filtType%=0 then		'If Low pass filter selected
		Print("var eFreq1 := %f;\t'Edge frequency\n",cFreq1);		'Set edge frequency
		if filtName%=2 then		'If Chebyshev filter, use ripple value
			Print("\nIIRLp(0,eFreq1,order%,filtName%,ripple);\t'Generated Low Pass filter");
		else
			Print("\nIIRLp(0,eFreq1,order%,filtName%);\t'Generated Low Pass filter");
		endif;
	
	case filtType%=1 then		'If High pass filter selected
		Print("var eFreq1 := %f;\t'Edge frequency\n",cFreq1);
		if filtName%=2 then		'If Chebyshev filter specified, use ripple value
			Print("\nIIRHp(0,eFreq1,order%,filtName%,ripple);\t'Generated High Pass filter");
		else
			Print("\nIIRHp(0,eFreq1,order%,filtName%);\t'Generated High Pass filter");
		endif;
	
	case filtType%=2 then		'If Band pass selected
		Print("var eFreq1 := %f;\t'Edge frequency(Low Pass)\n",cFreq1);	'Specify both edges
		Print("var eFreq2 := %f;\t'Edge frequency(High Pass)\n",cFreq2);
		if filtName%=2 then		'If Chebyshev filter selected, use ripple value
			Print("\nIIRBp(0,eFreq1,eFreq2,order,filtName%,ripple);\t'Generated Band Pass filter");
		else
			Print("\nIIRBp(0,eFreq1,eFreq2,order,filtName%);\t'Generated Band Pass filter");
		endif;

	case filtType%=3 then		'If Band stop selected
		Print("var eFreq1 := %f;\t'Edge frequency(Low Pass)\n",cFreq1);	'Specify both edges
		Print("var eFreq2 := %f;\t'Edge frequency(High Pass)\n",cFreq2);
		if filtName%=2 then		'If Chebyshev filter selected
			Print("\nIIRBs(0,eFreq1,eFreq2,order,filtName%,ripple);\t'Generated Band Stop filter");
		else
			Print("\nIIRBs(0,eFreq1,eFreq2,order,filtName%);\t'Generated Band Stop filter");
		endif;

endcase;
RestoreAll();
FrontView(script%);
return 0;
end;

'******************************************************************************************************
'						Procedures to hide and show script windows and toolbars
'******************************************************************************************************

proc HideAll()
var i%;
gFloat%[0] := App(-2);        'number of windows
for i% := 1 to gFloat%[0] do  'hide all windows and save state
   gFloat%[i%] := View(App(i%)).WindowVisible(0);
   next;
end

proc RestoreAll()	'Restore toolbar and script window
var i%;
for i% := 1 to gFloat%[0] do  'restore hidden windows
   View(App(i%)).WindowVisible(gFloat%[i%]);
   next;
end

the generated filters for a critical purpose. For a more in-depth description of the use of the IIR 
script commands, and the different types of filter, see the Spike2 on-line help.  

 

Scripts: Signal  

Q. I would like a script that allows me to print a specified range of frames from a data file. i.e., all 
Tagged frames, or a frame specifier such as 2, 4, 8 or 2...8. 

A. The attached script, PrintFrames.sgs, will print the specified frames and time range selected 
by the user. When the script is run, it prompts you to open a file and set up the view ready to print. 
You can then specify which frames to include. When you click OK, the selected frames are printed 
to the currently selected printer. 

Did you know…? 
You can now copy the position of horizontal cursors, and the differences in Y-axis units between multiple 
horizontal cursors to the clipboard. To do this, right-click a horizontal cursor and select Copy position from 
the context menu. 

Recent questions  
Q. Is there any way to save the values displayed by using the Alt and drag interactive 

measurements function? 

A. The interactive measurements feature in Spike2 now allows you to save the measurement values. 
 When dragging the cursor, release the Alt key and use either the C or L key to copy or log the 
 displayed values. 

   

User Group  

If you have any comments about the newsletter format and content, or wish to un-subscribe from the 
mailing list for this newsletter, please notify sales@ced.co.uk. 

mailto:sales@ced.co.uk

'$PrintFrames|Allows the user to print specified frames from a data file

var dataFile%;			'Handle of data file
var ok%;					'Returns whether user pressed OK
var list%[100];		'Array to hold frames
var noFrames$;			'String to hold specified frames
var c%;					'Variable acts as counter for printing
var useSpec%;			'Returns checkbox value in dialog
var frameSelect%;		'Returns selected frames
var frameSelect$[5];	'List for dialog
var state%;				'Sets frame state to include (if Frame State selected);

frameSelect$[0]:= "All frames";			'This sets up the list to use in the dialog
frameSelect$[1]:= "Current frame";
frameSelect$[2]:= "Tagged frames";
frameSelect$[3]:= "Un-tagged frames";
frameSelect$[4]:= "Frame state = xxx";

dataFile%:= FileOpen("",0,1);		'Prompts user to open a data file

Interact("Select time range to print. Then press OK",1023);		'Let user set time range and axis etc.
DlgCreate("Select Frames to Print");		'Creates a dialg letting the user select the frames to print
DlgList(1,"Frames",frameSelect$[]);			'List of frame types
DlgInteger(2,"Frame State",0,100);			'Field for frame state
DlgCheck(3,"Use frame specifier");			'Checkbox
DlgString(4,"Frames to print",20);			'Frame specification field
DlgAllow(0,0,DlgSet%);						'This enables dialog update to only enable relevant items in the dialog
ok%:= DlgShow(frameSelect%,state%,useSpec%,noFrames$);	'Return variables holding dialog field values
if ok% < 1 then Halt;endif;

if useSpec%=1 then			'If user has selected Frame specifier...
	FrameList(list%[],noFrames$);		'...Use specified frames for print
else
	docase
		case frameSelect%= 0 then	'If All frames selected..
			frameSelect%:= -1;		'..set All frames

		case frameSelect%=1 then	'If Curent frame selected..
			frameSelect%:= -2;		'..set Current frame

		case frameSelect%=2 then	'If Tagged frames selected..
			frameSelect%:= -3			'..set Tagged frames

		case frameSelect%=3 then	'If Un-tagged frames selected..
			frameSelect%:= -6			'..set Un-tagged frames
		
	endcase
FrameList(list%[],frameSelect%);	'Use selected frames for print
endif;

if frameSelect%=4 then		'If user selected Frame State
	FrameList(list%[],-1,state%);		'Only include frames with specified state
endif;

for c%:=1 to list%[0] do	'Loop through list of frames
	Frame(list%[c%]);			'Select frame
	FilePrintVisible();		'..and print
next;


Func DlgSet%(item%)	'This function is called when the dialog is updated
var v%,s%;			'variable to monitor dialog values
v%:= DlgValue(3);		'Value of Use Specifier checkbox (checked or not)
s%:= DlgValue(1);		'Value of Frame list
DlgEnable(1,1);		'Set default: Enable frame list..
DlgEnable(0,4);		'..Disable Specifier field..
DlgEnable(0,2);		'..Disable State field

if v% = 1 then
	DlgEnable(0,1);	'If checkbox is checked..
	DlgEnable(0,2);	'..disable Frame list..
	DlgEnable(1,4);	'..Enable specifier
endif;

if s%= 4 and v% <> 1 then	'If Frame state is selected and checkbox is clear
	DlgEnable(1,2);	'Enable State field
endif;
return 1;
end;

